
クラスの定義

• 財布クラスを作るにはどうする？

–内部的に持つ情報は？ 必要なメソッドは？

• 人情報を管理するクラスを作るにはどうす
る？

–姓，名，年齢

• ペイントツールで1つずつのストロークを管
理するには？

• 文字列を扱うにはどうするか？

文字列型

• String 型は，文字列を扱うためのクラス

–「中村聡史」「明治大学 総合数理学部」

• 文字列を扱うにはどういった機能が必要？

–文字列の長さを取得する

–文字列にある文字が含まれているかを調べる

–文字列を部分的に置き換える

–文字列が一致しているか調べる

– n文字目の文字を取得する

–などなど

Stringクラスのメソッド

• charAt( num ); num文字目の文字を返す（0から始まる）

• indexOf( 文字列 ); 入力された文字列が何文字目か？

• length(); 入力された文字の文字数を返す

• substring( x ); x文字目から最後までを出力

• substring( x, y ); x文字目からy-1文字目までを出力

• toLowerCase(); 全てを小文字に変換する

• toUpperCase(); 全てを大文字に変換する

• replace( 文字列A, 文字列B );

– 文字列Aを文字列Bに変更する

• split( 文字列 ); 文字列を分割

http://processing.org/reference/String.html
http://docs.oracle.com/javase/6/docs/api/java/lang/String.html

String str = "Department of Frontier Media Science (FMS), IMS, Meiji University";

println( str.length() );
println( str.charAt( 11 ) );
println( str.indexOf("F") );
println( str.indexOf("S") );
println( str.indexOf("Meiji") );
println( str.substring( str.indexOf("Meiji") ) );
println( str.toLowerCase() );
println( str.toUpperCase() );

65
o
14
29
49
Meiji University
department of frontier media science (fms), ims, meiji university
DEPARTMENT OF FRONTIER MEDIA SCIENCE (FMS), IMS, MEIJI UNIVERSITY

String [] ret = str.split( " " );
println( ret.length ) ;
int i=0;
while(i < ret.length)
{

println( ret[i] );
i++;

}

問題

• 下記の結果は？

println( str.indexOf("Media") );

println( str.toLowerCase().indexOf("media") );

println( str.substring(

str.indexOf("Meiji") ).length() );

• Frontier Media Science を出力するには？

str.substring(             ,               );

動物園の動物クラス群

• 猫，犬，猿，象，熊を定義
– それぞれの座標は意識したくない

• cat.x, cat.y, dog.x, dog.y, monkey.x, monkey.y, ...

• 内部で適当に処理してもらう

– 描画はシンプルにしたい

• cat.draw(), dog.draw(), monkey.draw(), elephant.draw(), ...

– 移動もシンプルにしたい

• cat.move(), dog.move(), monkey.move(), elephant.move(), ...

– 睡眠も任せてしまう

• cat.sleep(), dog.sleep(), monkey.sleep(), elephant.sleep(), ...

いいんだけど，重複があるのでは？

×が動き回るクラスを作ろう

Ballクラスを利用して、Crossクラスを作ろう！

– Ballクラスと、Crossクラスの違いは、表示される図形
が「〇」か「×」かなだけ！！

– Crossクラスのタブを作成し、Ballクラスをコピペ！

– BallをCrossに書き換える！

– 表示だけを変更したいので、displayの中身を変更する

– メインのプログラムで Cross を使っていこう！

(           ,           )

(            ,            )

(            ,            )

(            ,            )

(            ,            )

Ballを改良しCrossを作る
class Cross{

int x;
int y;
int speedX;
int speedY;

Cross(){
init();

}
void init(){
x = (int)random(width);
y = (int)random(height);
speedX = (int)random(5);
speedY = (int)random(5);

}
void display(){
line( x-15, y-15, x+15, y+15 );
line( x+15, y-15, x-15, y+15 );

}

void move(){
x = x + speedX;
y = y + speedY;
if ( x+15 > width ) {
x = width - 15;
speedX = -speedX;

}
if( x - 15 < 0 ){
x = 15;
speedX = -speedX;

}
if( y + 15 > height ){
y = height - 15;
speedY = -speedY;

}
if( y - 15 < 0 ){
y = 15;
speedY = -speedY;

}
}

}

Crossクラスを使うと
Cross fukuchi;
Cross nakamura;
Cross hashimoto;
void setup() {

size( 400, 300 );
fukuchi = new Cross();
nakamura = new Cross();
hashimoto = new Cross();

}

void draw() {
background(255);
fukuchi.move();
nakamura.move();
hashimoto.move();
fukuchi.display();
nakamura.display();
hashimoto.display();

}

ここで・・・

• Ballクラスと，Crossクラスはほとんど一緒

• 違いはクラス名とコンストラクタ，そして
displayのインスタンスメソッドのみ

Ball Cross

インスタンス変数 x, y
speedX, speedY

x, y
speedX, speedY

一致

クラス名 Ball Cross 違う

コンストラクタ Ball() Cross() 内部は一致

void init() 一致

void move() 一致

void display() fill(0, 0, 255);
ellipse(x, y, 30, 30);

line(x-15, y-15, x+15, y+15);
line(x+15, y-15, x-15, y+15);

違う

無駄じゃね？

前回の課題

• Ball クラスを改良し，×が動き回るCrossクラスと△が動
き回るTriangleクラスを作成せよ．

• またこれを利用して， 5個の〇と，4個の×と，3個の△が
動き回るプログラムを作成せよ

• ただし，その速度は目視可能なものとせよ

• 1個程度動いていないものがあってもよい

• また，可能であれば〇は壁で跳ね返り，×と△は跳ね返らずに反対
側から出てくるようにせよ

• 前回クラスを使っている人は、そのまま個数だけ変更して提出して
もOK

ほとんど重複している！
違うのはdisplay()だけ！
もっと簡単にできないの？

どう無駄をなくす？

• そもそもBallクラスというのがダメなのでは？

– Objectクラスという名前にして、objectTypeなどの変数
を用意し、displayの時に切り替えては？

class Object{
int x;
int y;
int speedX;
int speedY;
int objectType;

void display(){
if( objectType == 0 ){
ellipse( x, y, 30, 30 );

} else if( objectType == 1 ){
rect( x, y, 30, 30 ); 

}    
}

これも一つの方法
ただ複雑なものだと厳しい

そこで継承！

• 大辞林 第三版

1. 先の人の身分・権利・義務・財産などを受け継ぐこと
。 「王位を－する」

2. インヘリタンス→（オブジェクト指向プログラミング
において，クラス間でデータの共有を行う機構。新し
く定義するクラスを既存のクラスの下位クラスとして
記述し，上位クラスより属性やメソッドを引き継ぐ仕
組みをいう。上位クラスに対する差分のみを記述する
だけで新しいクラスを定義することが可能となる。）

一緒の部分をまとめた
スーパークラス（親クラス）を作る

スーパーなObjectクラス
class Object{

int x;
int y;
int speedX;
int speedY;

Object(){
init();

}
void init(){
x = (int)random(width);
y = (int)random(height);
speedX = (int)random(5);
speedY = (int)random(5);

}

void move(){
x = x + speedX;
y = y + speedY;
if ( x+15 > width ) {
x = width - 15;
speedX = -speedX;

}
if( x - 15 < 0 ){
x = 15;
speedX = -speedX;

}
if( y + 15 > height ){
y = height - 15;
speedY = -speedY;

}
if( y - 15 < 0 ){
y = 15;
speedY = -speedY;

}
}

}

display() は内容が違うので
削除してしまう

一緒の部分をまとめる

• Ball は Object の変数（x, y, speedX, speedY）
や機能（移動や初期化）をもち，独自の表示に関
する機能（メソッド）をもつクラス

• Cross はObject の変数（x, y, speedX, speedY）
や機能（移動や初期化）をもち，独自の表示に関
する機能（メソッド）をもつクラス

• インスタンス変数や，インスタンスメソッドを引
き継ぐことを継承と呼ぶ！

Objectクラスを使うと

class Ball extends Object 
{

void display(){
fill( 255, 0, 0 );
ellipse( x, y, 30, 30 );

}
}

class Cross extends Object 
{

void display(){
line( x-15, y-15, x+15, y+15 );
line( x+15, y-15, x-15, y+15 );

}
}

BallクラスとCrossク
ラスが劇的に短く！

Ball miyashita;
Ball komatsu;
Ball kikuchi;
Cross fukuchi;
Cross nakamura;
Cross hashimoto;

void setup() {
size( 400, 300 );
fill( 255, 0, 0 );
miyashita = new Ball();
komatsu = new Ball();
kikuchi = new Ball();
fukuchi = new Cross();
nakamura = new Cross();
hashimoto = new Cross();

}
：

なぜ動くの？

• 継承すると，親の力をすべて引き継ぐ！

• 継承の方法は extends とやるだけ！

• 継承により親の能力，値はすべて引き継ぎ
ます

class クラス名 extends 親クラス名 {

}

じゃあ

• Object クラスを継承して「三角形」を描画
するクラスを作るには？（Triangle）

• Object クラスを継承して「□」を描画する
クラスを作るには？（Square）

Triangle

display()

Object

x, y
speedX
speedY

init()

move()

Object()

三角形と四角形のクラス

• 動く三角形のクラスと動く四角形のクラス
class Triangle extends Object 
{
void display()
{
fill( 0, 255, 0 );
triangle( x, y-15, x-15, y+15, x+15, y+15 );

}
}

class Square extends Object 
{
void display()
{

fill( 0, 0, 255 );
rect( x-15, y-15, 30, 30 );

}
}

つまり先週の課題は

• Ball クラスを改良し，×が動き回るCrossクラスと
△が動き回るTriangleクラスを作成せよ．

• またこれを利用して， 5個の〇と，4個の×と，3個
の△が動き回るプログラムを作成せよ

• ただし，その速度は目視可能なものとせよ

• 1個程度動いていないものがあってもよい

• また，可能であれば〇は壁で跳ね返り，×と△は跳ね返
らずに反対側から出てくるようにせよ

• 前回クラスを使っている人は、そのまま個数だけ変更し
て提出してもOK

Objectクラスを継承してCross, Triangleクラ
スを作り，displayだけ独立させる！

三角形は跳ね返らないように

(Q3) 先述のObjectを継承したTriangleクラス
を改良し，三角形は跳ね返らず右端→左端，
左端→右端，上端→下端，下端→上端と移動
するようにせよ

• 考え方

– Object の move メソッドを Triangle クラス内
でオーバーライドして，三角形専用のメソッド
を作成する

クラス分析class Triangle extends Object {
void display(){
fill( 0, 255, 0 );
triangle( x, y-15, x-15, y+15, x+15, y+15 );

}
void move(){
x = x + speedX;
y = y + speedY;
if( x > width ){
x = x - width;

} else if( x < 0 ){
x = x + width;

}
if( y > height ){
y = y - height;

}
if( x < 0 ){
y = y + height;

}
}

}

moveメソッドをオーバーライドして
親の move メソッドが
呼ばれないようにする

move をオーバーライド

• Triangle の move で，Object の move を
上書きしてしまう！

Triangle

move()

display()

Object

x
y
speedX
speedY

init()

move()

Object()

全体の挙動を変更

(Q4) Objectクラスのmoveメソッドを変更し，
上端及び下端は跳ね返るが，左端と右端では
逆側から現れるようにせよ

• 考え方

– Object の move メソッドのみ変更

– y 座標の条件で跳ね返り

– x 座標の条件で逆側から現れるようにする

Object クラスのみ変更
class Object
{

int x;
int y;
int speedX;
int speedY;

Object()
{
init(); 

}

void init()
{
x = (int)random(width);
y = (int)random(height);
speedX = (int)random(5);
speedY = (int)random(5);

}

void move()
{
x = x + speedX;
y = y + speedY;
if( x > width ){
x = x - width;

} else if( x < 0 ){
x = width + x;

}

if( y + 15 > height ){
y = height - 15;
speedY = -speedY;

} else if( y - 15 < 0 ){
y = 15;
speedY = -speedY;

}
}

}

惑星と衛星の様なオブジェクト

(Q5) 800x600のウインドウ内で，任意の場所x，yか
ら任意の速度で移動する3つの赤色の円（直径30ピク
セルの惑星）を描画し，右端・左端・上端・下端に
来ると跳ね返るようにする．また，赤色の円（惑星
）には円の中心から50の距離があるところに1つの衛
星（直径10ピクセル）があり，5度ずつ円の周りを回
転するようにせよ

• 考え方

– 円の中心（x, y）から衛星の方向の角度（0～360度）を
theta とすると，衛星の座標は

(x+50*cos(radians(theta)), y+50*sin(radians(theta)))

惑星と衛星の様なオブジェクト

• Objectクラスを継承して，変数を追加する

class PlanetSatellite extends Object 
{

int theta;
int sx;
int sy;
void display()
{

fill( 255, 0, 0 );
ellipse( x, y, 30, 30 );
theta = theta + 5;
int sx = (int)(x+50*sin(radians(theta)));
int sy = (int)(y+50*cos(radians(theta)));
ellipse( sx, sy, 10, 10 );

}
}

惑星と衛星の様なオブジェクト

PlanetSatellite miyashita;
PlanetSatellite komatsu;
PlanetSatellite kikuchi;
void setup() 
{
size( 800, 600 );
miyashita = new PlanetSatellite();
komatsu = new PlanetSatellite();
kikuchi = new PlanetSatellite();

}

void draw() 
{
background(255);
miyashita.move();
komatsu.move();
kikuchi.move();
miyashita.display();
komatsu.display();
kikuchi.display();

}

継承すると・・・

• Object を PlanetSatellite として継承し，theta 
という変数と，display() というメソッドを追加

Planet
Satellite

theta

display()

Object

x
y
speedX
speedY

init()

move()

Object()

惑星と衛星の様なオブジェクト

(Q5) Q4を改良し，衛星の開始角を0～360度の任意の
場所にしたい．どうするか？

• 考え方

– void init() というインスタンスメソッドを追加し，改
良したらOK？

やってみる

class PlanetSatellite extends Object {

int theta;

void init(){

theta  = (int)random(360);

}

void display(){

fill( 255, 0, 0 );

ellipse( x, y, 30, 30 );

theta = theta + 5;

int sx = (int)(x+50*sin(radians(theta)));

int sy = (int)(y+50*cos(radians(theta)));

ellipse( sx, sy, 10, 10 );

}

}

init() はコンストラクタで
呼ばれるObjectクラス参照

画面の左上から動かないのはなぜ？

上書きしてしまったから

• Object の init() を，PlanetSatellite の
init() でオーバーライドている（Objectのinit()
が呼び出されない）

Planet
Satellite

theta

display()

Object

x
y
speedX
speedY

init()

move()

Object()

init()

やってみる

class PlanetSatellite extends Object {
int theta;
void init(){
theta  = (int)random(360);
x = (int)random(width);
y = (int)random(height);
speedX = (int)random(5);
speedY = (int)random(5);

}
void display(){
fill( 255, 0, 0 );
ellipse( x, y, 30, 30 );
theta = theta + 5;
int sx = (int)(x+50*sin(radians(theta)));
int sy = (int)(y+50*cos(radians(theta)));
ellipse( sx, sy, 10, 10 );

}
}

Object の init() にあるの
をそのままコピペする

動くけど，なんだか
無駄が増えている・・・

Objectのinit()も使いたい

• クラスの中で，super と書くと，継承元の
親を呼び出すことができる！

class PlanetSatellite extends Object 
{
int theta;
void init(){
theta  = (int)random(360);
super.init();

}
void display(){
fill( 255, 0, 0 );
ellipse( x, y, 30, 30 );
theta = theta + 5;
int sx = (int)(x+50*sin(radians(theta)));
int sy = (int)(y+50*cos(radians(theta)));
ellipse( sx, sy, 10, 10 );

}
}

super . メソッド名

課題3-1: PlanetSatellite2

• Q4の PlanetSatellite クラスを継承し，惑星の中
心から80ピクセル離れた位置に青色の小さな衛星
が2つある（直径5ピクセルで毎フレーム3度ずつ動
く） PlanetSatelliteEx クラスを作成せよ．また
，そのクラスを用いて4個の惑星（3つの衛星があ
る）がウインドウ内を動きまわるようにせよ

– 考え方

• インスタンス変数を追加する

• init メソッドと，
display メソッドを
オーバーライドする

課題3-2: boundAll3

• Objectクラスを継承し，赤色の丸が動き回るBall
クラスと，緑色の四角形が動き回るSquareクラス
，青色の三角形が動き回るTriangleクラスを作成
せよ

• また，この継承したクラスを利用して100個の赤色
丸と，80個の緑色四角形と，60個の青色△を動か
すようにせよ

課題3-3  DrawChara

• 配布するObjectCharaクラスを継承し、ObjectCharaクラス
内のcenterX, centerYを利用してその位置に自身のキャラ
クタを描画するようにせよ。なお、継承したクラスでは
displayとdisplayExをオーバーライドせよ。displayは先
週の宿題を利用し、displayExはそれを改良して色や姿勢
を変えるなどして区別できるよう工夫せよ（大きな改良は
不要）

• なおメインプログラムは、キャラクタが動き回りdisplay
で描画されるものであるが、マウスのボタンが押されてい
るときにはdisplayExが呼び出されて色違いが描画される
ことになる。

• クラスの名前は名前と番号が含まれるようにし、そのクラ
スのみ別ファイルとして作成せよ

– （例）1-2-33 中村の場合、クラス名を「ExNakamura33」とする

• 来週配布して利用します

宿題

• 100秒タイマークラス（TimerBase）を利用して残
り時間を表示してみよう

– ただし，タイマーはクリックの度にカウントダウンを開
始し，0秒になったら停止するようにせよ

– TimerBaseは講義資料サイトからダウンロードしよう

– また，100秒タイマークラスを継承し2つのタイマーを作
成せよ．1つはプログレスバーのスタイルで残り時間を
表示するもの，もう1つはなにか適当なものを考えて作
成せよ．


